



# 二次及三角函數圖形教學 的微電腦程式設計及其應用

林保平

在函數及其圖形性質的教學中，通常都須經由描點畫圖，來比較說明圖形與圖形間的異同及函數式中係數對圖形的影響。有些函數，計算困難不易求值；有些函數，函數值不是整數，描點困難；比較兩個函數圖形也不太容易。本文針對這些困難，特就二次函數及三角函數，提供了一個BASIC 程式，利用微電腦，作較快而且較準確的畫圖工作。只有係數不同的函數圖形，可畫在相同的坐標系中加以比較。學生利用這個程式，能夠便利地觀察比較函數圖形的形狀、位置，圖形的平移、疊合及其與函數係數的關係，並探究其原因，解釋其結果。

本程式是在統一PC 101 電腦上作出來的，可適用於APPLE II的相容機種。程式在執行之後，首先呈現下列目錄，以供選擇：

IRUN

```
*****
* GRAPH OF FUNCTION *
* BY PAUL PAO-PING LIN *
*****
```

CHOOSE ONE OF FOLLOWING :

- (1)  $Y=A*(X-B)^2+C$
- (2)  $Y=A*ABS(X-B)+C$
- (3)  $Y=D*SIN(A*(X-B))+C$
- (4)  $Y=D*TAN(A*(X-B))+C$
- (5)  $Y=D*SEC(A*(X-B))+C$
- (6)  $Y=D*SIN(A*X)+B*COS(C*X)$

(7)  $Y=D*SIN(A*X)+B*SIN(C*X)$

(8) TRANS PARABOLA


(9) TRANS ELLIPS

其中(1)至(7)為畫圖程式，給予適當的D、A、B、C值，可觀察並探討D、A、B、C等係數對函數圖形的影響。

(8)、(9)為驗證及熟悉拋物線與橢圓的平移及90度旋轉的程式。

若所選的是(1)至(7)中的一個，螢光幕上將出現一個直角坐標系，並標出x軸及y軸的標尺，如圖0.1所示。坐標軸下方第一列的文字是所欲畫圖的函數式，及x軸、y軸的標尺單位。如圖0.1所示， $Y=D*TAN(A*(X-B))+C$ 為函數式， $V=1$ 及 $H=PI$ 指縱軸單位為1，橫軸單位為 $\pi$ 。最下方一列印出的是“OK! THEN”，等待輸入指令。指令在函數式下方兩列。現解釋於下：

① D, A, B, C = CHANGE 表示，如果按D, A, B, C任一鍵，即可輸入此係數之值。例如按B，則螢幕“OK! THEN”的下方，將出現“FIX B=”，此時即可輸入B之值。若輸入B值為2，則第一列函數式中的B將被2取代，成為 $Y=D*TAN(A*(X-2))+C$ ，而“OK! THEN”又再出現，等候輸入其他指令；此時若再按B，則第一列函數式又回復原狀。依此，使用者可以定D, A, B,



```
Y= D *TAN( A *(X- B ))+ C  V=1 H=PI
[D; A; B; C=CHANGE, RTN=CONT, N=NEW, T=MENU ]
OK! THEN
```

圖 0.1 選用畫圖程式(4)時，螢幕所出現的畫面。

C，之值，亦可留下一兩個未知數不加定值，以觀察此係數對圖形的影響。

②  $N=NEW$  表示，若按  $N$ ，可清除畫面上原有的圖形。

③  $E=END$  表示，若按  $E$  鍵，程式終止。

④  $T=MENU$  表示，若按  $T$  鍵，則程式跳往目錄頁，可以重新選取某一類函數來作圖。

⑤  $RTN=CONT$  表示，若按 RETURN 鍵，則可繼續畫圖，不改變函數式，也不清除畫面。此鍵按下後，螢幕下方將出現“INPUT”字幕，此時即可輸入尚未定值的係數值，函數式中有幾個未知數，則依次鍵入幾個數值（必須整數或小數），每一數值之間以逗號分開


，完成後，按 RETURN 鍵，程式即開始作圖，完成圖形之後，“OK! THEN”出現，等候新指令。

若選(8)或(9)，螢幕上將出現一個平面直角坐標系，上面還有一個在標準位置的拋物線（選(8)時）或橢圓（選(9)時）。如圖 0.2(a)及 0.2(b)所示，第一列字幕所示者為拋物線的函數式  $Y=(X-O)^2+O$ （選(8)時）或橢圓方程式

$$\frac{(X-O)^2}{1600} + \frac{(Y-O)^2}{400} = 1 \quad (\text{選(9)時})。$$


第二列及第三列字幕列印出平行或鉛直移動圖形，或旋轉圖形之指令，說明如下：

①  $I=UP$  表示，按  $I$  鍵，圖形往上移一



```
FUNCTION Y = ( X- 0 )^2 + 0
[ I=UP, J=LEFT, K=RIGHT, M=DOWN, T=MENU ]
[ 1-9=STEP, H=C C.W, L=C.W, E=END, N=NEW ]
```

圖 0.2 (a) 選用(8) TRANS PARABOLA時螢幕所出現畫面，操作者可平移或旋轉此拋物線，函數式也會配合圖形之變化而改變。



```
ELLIPS (X- 0)^2/1600+(Y- 0)^2/ 400=1
[ I=UP,J=LEFT,K=RIGHT,M=DOWN,T=MENU ]
[ 1-9=STEP,H=C.C.W,L=C.C.W,E=END,N=NEW ]
```

圖 0.2 (b) 選用(9)TRANS ELLIPS 時螢幕所出現畫面，操作者可平移或旋轉此橢圓，方程式亦將配合圖形變化而改變。

步 (STEP)。

②  $J = LEFT$  表示，按  $J$  鍵，圖形往左移一步。

③  $K = RIGHT$  表示，按  $K$  鍵，圖形往右移一步。

④  $M = DOWN$  表示，按  $M$  鍵，圖形向下移一步。

⑤  $1-9 = STEP$  表示，按 1 至 9 的任一數字鍵，可指定一步 (STEP) 的步幅為幾單位。1 至 9 依次為  $0.1/2$  到  $0.9/2$  單位 (選 (8) 時) 或 1 到 9 單位 (選 (9) 時)。程式預定的步幅為每步 0.1 單位 (選 (8) 時) 或 2 單位 (選 (9) 時)。

⑥  $H = C.C.W$  表示，按  $H$  鍵，圖形將依逆時鐘方向旋轉 90 度。

⑦  $L = C.C.W$  表示，按  $L$  鍵，圖形將依順時鐘方向旋轉 90 度。

⑧  $N = NEW$  表示，按  $N$  鍵，圖形將回到標準位置。

⑨  $T = MENU$ ， $E = END$  之意義與前面所述相同。

在圖形移轉過程中，第一列字幕中的函數式或方程式，將因應圖形位置的改變而顯示出正確的式子。例如，橢圓之方程式原為

$$\frac{(X-0)^2}{1600} + \frac{(Y-0)^2}{400} = 1, \text{ 若原地依順時鐘}$$

方向旋轉 90 度 (即按  $L$  鍵)，則方程式變為

$$\frac{(X-0)^2}{400} + \frac{(Y-0)^2}{1600} = 1. \text{ 若右移 5 步,}$$

$$\text{上移 10 步, 則方程式為 } \frac{(X-10)^2}{1600} + \frac{(Y-20)^2}{400}$$

$$= 1.$$

在教學時，教師可依課程內容，設計一些循序發展的問題發給學生，由教師依問題之次序，配以適當的問答演示給學生看 (只須一部電腦) 或將學生分組，分發預先準備的磁片拷貝，讓學生自行依問題的次序利用電腦畫圖，研究圖形的形狀位置，回答問題，教師從旁指導，然後由各組交換分享學習的成果。(須要多部電腦)。

下面列出一些實驗及討論性的問題和說明，供讀者參考，這些問題是參酌師專數學第一冊第 5-1 節二次函數，第二冊第 1-3 節三角函數之週期及圖形，以及第 3-3 節波動的疊合原理而設計的。

### 一、關於函數 $y = A(x-B)^2 + C$ 之問題

(1) 試就不同的  $A$  值，畫出  $y = Ax^2$  之圖形，並比較說明  $A$  值對圖形的影響。

(2) 試就不同的  $B$  值，畫出  $y = (x-B)^2$  的圖形，並比較說明  $B$  值對圖形的影響。同樣，再觀察  $C$  對  $y = x^2 + C$  圖形之影響。

／前述二題，希望學生經由程式繪圖，發




圖 1.1  $y = Ax^2$  之圖形，其中  $A = \pm 0.1, \pm 0.3, \pm 0.5, \pm 1$ ， $A$  正時，開口向上； $A$  負時，開口向下， $|A|$  愈小，開口愈大。




圖 1.2 (a)  $y = (x - B)^2$  之圖形，其中  $B = 1, 3, 0, -3, -1$  拋物線與  $x$  軸的交點橫坐標為  $B$  值。




圖 1.2 (b)  $y = x^2 + C$  之圖形，其中  $C = 3, 2, 1, 0, -1, -2, -3$  拋物線與  $y$  軸交點的縱坐標為  $C$  值。

現  $A, B, C$  值分別與拋物線圖形的開口大小、開口之方向，圖形的水平鉛直移位有關。參看圖 1.1、1.2(a)、1.2(b)。

(3) 對不同的  $A, B, C$  值，觀察  $y = A(x - B)^2 + C$  之圖形與  $y = x^2$  圖形的關係。

// 本題希望學生整體地看係數與圖形的關係，並了解  $A, B, C$  之值對圖形並沒有交叉的影響。

(4) 是否所有的二次函數均可化為  $Y = A(X - B)^2 + C$  之形式？

$Y = A(X - B)^2 + C$  之形式？

(5) 若在式子  $Y = A(X - B)^2 + C$  中，令  $Y - C = Y'$ ， $X - B = X'$ 。原式可化為怎樣的方程式？

// 教師可趁此時，介紹平移的概念，引導學生注意到平移的功能。欲研究二次函數圖形之性質，只須考慮  $Y = AX^2$  形式之二次圖形即可。教師亦可選用程式(8)，TRANS PARABOLA來增強學習結果。

(6)課本上，我們所見拋物線，均為開口向上或向下，若有拋物線開口向左或向右，它們是函數圖形嗎？它們的方程式應如何？

／本題假設學生未曾學習二次曲線。教師可提示學生將圖形旋轉 90 度來觀察，或將  $x$  軸改為  $y$  軸， $y$  軸改為  $x$  軸，所得圖形即為開口向左或向右。(由於拋物線的對稱性，故旋轉 90 度亦可。)利用 *TRANS PARABOLA* 程式可以很方便地依順時鐘或逆時鐘方向來旋

轉圖形。

## 二、關於函數 $y = D \sin A(x - B) + C$ 之利用

除了與前述問題一樣，分別探究  $D$ ， $A$ ， $B$ ， $C$  四係數對函數圖形之振幅、週期、水平移動、鉛直移動等關係外(參看圖 2.0(a)、2.0(b)、2.0(c))，下列問題可供參考。




圖 2.0(a)  $y = 3 \sin(x - B)$  之圖形  
 ，其中  $B = 0, 1, 1.75(\frac{\pi}{2}), 2, 3.14(\pi), 6.28(2\pi)$   
 $B$  之作用為將  $y = 3 \sin x$  之圖形作水平移動  $|B|$  單位， $B > 0$  右移， $B < 0$  左移， $B = 0$  及  $B = 6.28$  時圖形重疊，亦即驗正了  $2\pi(6.28)$  為週期。




圖 2.0(b)  $y = D \sin x$  之圖形，其中  $D = -0.5, -1, -3, -5$ 。  
 $y = D \sin x$  圖形之振幅為  $y = \sin x$  振幅之  $|D|$  倍， $D < 0$  使  $y = D \sin x$  之圖形與  $y = |D| \sin x$  之圖形函數值符號相反。




圖 2.0(c)  $y = \sin x + C$  之圖形，其中  $C = -5, -3, -1, 1, 3, 5$ 。 $C$  之作用為將  $y = \sin x$  之圖形鉛直移動  $|C|$  單位， $C < 0$  下移， $C > 0$  上移。

(1)依下表所列  $A$  值繪圖，並估計  $y = 4 \sin Ax$  之週期。

$A$ 值	$\frac{4}{5}$ (0.8)	$\frac{3}{4}$ (0.75)	$\frac{2}{3}$ (0.67)	$\frac{1}{2}$ (0.5)	$\frac{1}{3}$ (0.33)	1	2	3	4
週期			$3\pi$			$2\pi$			

若表中  $A$  值為負時，情形如何？




圖 2.1  $y = 4 \sin Ax$  之圖形，其中  $A = 0.75, 0.67, 0.5, 0.33, 1, 4$ ， $y = 4 \sin Ax$  的週期變為  $y = 4 \sin x$  週期的  $\frac{1}{|A|}$  即  $\frac{2\pi}{|A|}$ 。

(2)就下列各組  $D, A, B$  之值觀察函數  $Y = D \sin A(X - B)$  之圖形，並解釋其原因

$D$	-1	-1	1	1	1	1
$A$	1	1	1	1	-1	1
$B$	$6.28(2\pi)$	0	$3.14(\pi)$	$6.28$	$3.14$	0

類似的問題，可讓學生探討簡單的三角恒等式，如  $\sin(x - \pi) = -\sin(\pi - x) = -\sin x$ ， $\sin(x - 2\pi) = \sin x$ ， $\sin[-(x - \pi)] = \sin x$ ……等等。

(3)定出  $D, A, B, C$  之值，利用  $y = D \sin A(x - B) + C$  畫出  $y = \cos x$  之圖形，這些值是否唯一確定？試舉出某些可能的情况。

本題讓學生了解或發現正弦函數與餘弦函數間的關係。將正弦圖形左移  $2n\pi + \frac{\pi}{2}$  或

右移  $2n\pi + \frac{3\pi}{2}$ ， $n \in \{0\} \cup N$ ，即可得到餘弦函數的圖形。

(4)定出  $D, A, B, C$  之值，利用  $y = \sin A(x - B) + C$  之程式，畫出  $y = 4 \sin^2 x$

本題學生可探究  $A$  值與  $y = 4 \sin Ax$  週期之關係，教師宜引導學生用大尺度方法，觀察圖形的週期（例如  $8\pi$  中有 3 個重複圖形，故週期為  $\frac{8\pi}{3}$ ）並發現  $y = 4 \sin Ax$  之週期為  $\frac{2\pi}{|A|}$ ，而振幅 4 對週期並無影響（此處用 4 是因其圖形較清楚。）。參看圖 2.1

及  $y = 4 \cos^2 x$  之圖形，並觀察其週期及振幅。

本題及類似的問題，可讓學生探討二倍角有關的恒等式，學生可將函數式  $y = 4 \sin^2 x$  及  $y = 4 \cos^2 x$  化為形如  $y = D \sin A(x - B) + C$  之形式，求出  $D, A, B, C$  之一組值。例如， $y = 4 \sin^2 x = 4 \cdot \frac{1 - \cos 2x}{2} = 2 - 2 \cos 2x = 2 - 2 \sin(\frac{\pi}{2} - 2x) = (-2) \sin[(-2)(x - \frac{\pi}{4})] + 2$ ，故取  $D = -2, A = -2, B = \frac{\pi}{4}, C = 2$  即可畫出  $y = 4 \sin^2 x$  之圖形，其週期為  $\pi$ ，振幅為 2。 $D, A, B, C$  之值並非唯一。

(5)可否用  $Y = D \sin A(X - B) + C$  之程式，畫出  $y = E \sin^2 x + F$  及  $y = E \cos^2 x + F$  之圖形？其週期如何？振幅如何？

依上題類似的推論，學生很容易便可確定，此類函數週期為  $\pi$ ，振幅為  $|\frac{E}{2}|$ 。

### 三、關於函數 $y = D \sin Ax + B \cos Cx$ 之利用

本程式可用來探討波動的疊合原理。教師可先簡略說明或由學生自行研究兩函數  $f, g$  之圖形和函數  $h = f + g$  圖形的關係—— $h$  之圖形每一點的  $y$  坐標為對應之  $f$  及  $g$  圖形上兩點的  $y$  坐標的和。此外下面的問題可供參考。

(1) 利用  $y = D \sin Ax + B \cos Cx$  程式，畫圖觀察函數  $y = D \sin x + D \cos x$  之圖形，它是否週期函數，其形狀如何？圖形的最高點及最低點如何？（即振幅）

／本題希望學生能定出不同  $D$  值，觀察圖形之變化，看出  $D$  對振幅的影響及平移的特性，作為探究  $D \sin x + D \cos x = D\sqrt{2} \sin(x + \frac{\pi}{4}) = D\sqrt{2} \cos(x - \frac{\pi}{4})$  之關係式的動機，並據此確定，此函數圖形之週期、振幅與正弦或餘弦圖形的關係。參看圖 3.1。




圖 3.1  $y = 3 \sin x, y = 3 \cos x$  及  $y = 3 \sin x + 3 \cos x$  之圖形，三者形狀類似，後者振幅不同。




圖 3.2  $y = D \sin x + B \cos x$  之圖形，其中  $(D, B) = (0.5, 1), (1, 2), (2, 4), (3, 6)$ 。其圖形的形狀與  $y = D \sin x + D \cos x$  相似，只是振幅不同。

(2) 就不同的  $D, B$  值，畫圖觀察  $y = D \sin x + B \cos x$  及  $y = D \sin x, y = B \cos x$  圖形間的關係。

／本題立在上題的基礎上，鼓勵學生自行探究關係式  $y = D \sin x + B \cos x = \sqrt{D^2 + B^2} \sin(x + \alpha) = \sqrt{D^2 + B^2} \cos(x - \beta)$ ，其中  $\tan \alpha = \frac{B}{D}, \tan \beta = \frac{D}{B}$ 。參看圖 3.2。

(3) 就不同的  $A$  值，畫圖觀察  $y = \sin Ax + \cos Ax$  之圖形，其週期如何？

／本題希望學生發現  $y = \sin Ax + \cos Ax$  之週期與  $y = \sin Ax$ （或  $y = \cos Ax$ ）之週期均為  $\frac{2\pi}{|A|}$ 。參看圖 3.3。

(4) 就下列  $A, C$  值，畫圖觀察  $y = 2 \sin Ax + 2 \cos Cx$  之週期，看它是否有某一規律性。

A	2	2	3	3	5	6	12	1
C	3	4	6	9	3	8	7	2
週期			$\frac{2\pi}{3}$					

依左表所列  $A, C$  值畫圖，教師可以引導學生歸納出下列結論“若  $A, C$  為整數，且  $[A, C]$  表  $A, C$  之最大公因數，則函數  $y = \sin Ax + \cos Cx$  的週期為  $\frac{2\pi}{[A, C]}$ ”，並設法證明它。參看圖 3.4 (a), 3.4 (b)。

(5) 設  $A, C$  為有理數，則  $y = \sin Ax + \cos Cx$  是否週期函數，其週期如何？




圖 3.3  $y = \sin Ax + \cos Ax$  之圖形，其中  $A = 0.5, 1, 2$ 。其週期為  $\frac{2\pi}{|A|}$ ，與  $y = \sin Ax$  (或  $y = \cos Ax$ ) 之週期相同。




圖 3.4 (a)  $y = 2 \sin 2x + 2 \cos 3x$  之圖形，其週期為  $2\pi$ 。




圖 3.4 (b)  $y = 2 \sin 3x + 2 \cos 9x$  之圖形，其週期為  $\frac{2\pi}{3}$ 。

此處可用大尺度之方法觀察，在  $8\pi$  中有 12 個重複圖形。圖形不夠平滑是因為螢幕解析度不夠大的緣故。



∥本題希望學生自行依上題之方法列表，  
 探討函數之週期。設  $A = \frac{n_1}{m_1}$ ， $C = \frac{n_2}{m_2}$ ，其中

$n_1, m_1$  互質， $n_2, m_2$  互質，則函數  
 $y = \sin Ax + \cos Cx$  之週期為  $\frac{2\pi(m_1, m_2)}{[n_1, n_2]}$

，其中  $(m_1, m_2)$  指  $m_1, m_2$  的最小公倍數。  
 教師宜利用前題之結果，給學生充分的提示。

(6)定出  $D, A, B, C$  之值，利用  $y = D \sin Ax + B \sin Cx$  之程式，畫出  $y = 4 \sin^3 x$  之圖形，其週期如何？可否畫出  $y = 4 \cos^3 x$  之圖形？何故？




圖 4.1  $Y = D \tan A(x-B) + C$  之圖形，其中  $A = 1, B = 0, C = 0, D = 0.5, 1, 2, 5$ 。




圖 4.2  $Y = D \sec A(x-B) + C$  之圖形，其中  $A = 1, B = 0, D = 1, C = -4, -1, 0, 1, 3$ 。




圖 4.3  $Y = D \tan A(x-B) + C$  之圖形，其中  $A = 1, B = 0, D = 1, C = -5, -2, 0, 2, 3, 4$ 。

利用三倍角公式  $y = 4 \sin^3 x = 3 \sin x - \sin 3x$ ，取  $D = 3$ ， $A = 1$ ， $B = -1$ ， $C = 3$  即可，其週期為  $2\pi$ 。但  $y = 4 \cos^3 x = \cos 3x + 3 \cos x = \sin\left(\frac{\pi}{2} - 3x\right) + 3 \cos x = \sin\left(-3\left(x - \frac{\pi}{6}\right)\right) + 3 \cos x$ ，由於  $\frac{\pi}{2}$  及  $\frac{\pi}{6}$  這兩個常數，故無法由  $y = D \sin Ax + B \sin Cx$  或  $y = D \sin Ax + B \cos Cx$  之程式作圖。若有  $y = D \cos Ax + B \cos Cx$  或  $y = D \sin A(x - B)$

$+ C \cos Ex$  之程式即可。

其他函數之繪圖應用與前述三種，大同小異，不再贅述。參看圖 4.1 ~ 4.3。其實對任意函數  $y = f(x)$  而言， $Y = Df(A(x - B)) + C$  之圖形與係數之關係均相同。 $|D|$  或  $|A|$  之功能是将  $y = f(x)$  之圖形，對  $x$  軸或  $y$  軸作一定比例的壓縮或擴張； $B$ ， $C$  之功能是将  $y = f(x)$  之圖形，作水平或鉛直的移動（即平移）。

最後，我們來看看程式本身（如附表所示）。

JFOKE 1657,80

IRUN 5000

```

2 TEXT : HOME :L = 8
5 PRINT SPC(L + 1) "*****"
7 PRINT SPC(L + 1) "* GRAPH OF FUNCTION *"
8 PRINT SPC(L + 1) "* BY PAUL PAO-PING LIN *"
9 PRINT SPC(L + 1) "*****"
10 PRINT : PRINT SPC(L) : PRINT "CHOOSE ONE OF FOLLOWING : "
11 PRINT : PRINT SPC(L) : PRINT " (1) Y=A*(X-B)^2+C "
12 PRINT : PRINT SPC(L) : PRINT " (2) Y=A*ABS(X-B)+C"
13 PRINT : PRINT SPC(L) : PRINT " (3) Y=D*SIN(A*(X-B))+C"
14 PRINT : PRINT SPC(L) : PRINT " (4) Y=D*TAN(A*(X-B))+C"
15 PRINT : PRINT SPC(L) : PRINT " (5) Y=D*SEC(A*(X-B))+C"
16 PRINT : PRINT SPC(L) : PRINT " (6) Y=D*SIN(A*X)+B*COS(C*X) "
17 PRINT : PRINT SPC(L) : PRINT " (7) Y=D*SIN(A*X)+B*SIN(C*X) "
18 PRINT : PRINT SPC(L) : PRINT " (8) TRANS PARABOLA "
19 PRINT : PRINT SPC(L) : PRINT " (9) TRANS ELLIPS ";
20 GET TS
23 B = 1:C = 1:A = 0:A$ = " A " :D = 0:B$ = " B " :C$ = " C " :D$ = " D "
25 HGR : HCOLOR= 3
30 HPLOT 140,0 TO 140,159
40 HPLOT 0,80 TO 279,80
41 IF TS > 7 THEN TS = TS - 1: GOTO 1000
42 HOME : VTAB 21
45 IF TS < 3 THEN GOSUB 285
50 IF TS > 2 THEN GOSUB 420
52 PRINT "[D;A;B;C=CHANGE,RTN=CONT,N=NEW,T=MENU ]"
55 POKE 34,22
57 GOTO 150
60 PRINT "INPUT ";: IF TT = 7 THEN INPUT " ";G(1),G(2),G(3)
62 IF TT = 1 THEN INPUT " ";G(3)
63 IF TT = 2 THEN INPUT " ";G(2)
64 IF TT = 4 THEN INPUT " ";G(1)
65 IF TT = 3 THEN INPUT " ";G(2),G(3)
67 IF TT = 5 THEN INPUT " ";G(1),G(3)
69 IF TT = 6 THEN INPUT " ";G(1),G(2)
70 IF TT = 8 THEN INPUT " ";G(4)
71 IF TT = 9 THEN INPUT " ";G(4)
73 IF TT = 10 THEN INPUT " ";G(4),G(2)
74 IF TT = 11 THEN INPUT " ";G(4),G(2),G(3)
75 IF TT = 12 THEN INPUT " ";G(4),G(1)
76 IF TT = 13 THEN INPUT " ";G(4),G(1),G(3)
77 IF TT = 14 THEN INPUT " ";G(4),G(1),G(2)
78 IF TT = 15 THEN INPUT " ";G(4),G(1),G(2),G(3)
80 IF TS = 1 THEN DEF FN F(X) = (G(1) * (X - G(2)) ^ 2 + G(3)) * 20

```

```

81 IF TS = 3 THEN DEF FN F(X) = YS * (G(4) * SIN (G(1) * (X - G(2))) +
G(3))
82 IF TS = 4 THEN DEF FN F(X) = YS * (G(4) * TAN (G(1) * (X - G(2))) +
G(3))
83 IF TS = 5 THEN DEF FN F(X) = YS * (G(4) / COS (G(1) * (X - G(2))) +
G(3))
84 IF TS = 6 THEN DEF FN F(X) = YS * (G(4) * SIN (G(1) * X) + G(2) *
COS (G(3) * X))
85 IF TS = 2 THEN DEF FN F(X) = (G(1) * ABS (X - G(2)) + G(3)) * 20
86 IF TS = 7 THEN DEF FN F(X) = YS * (G(4) * SIN (G(1) * X) + G(2) *
SIN (G(3) * X))
87 S = 0
88 FOR X1 = - 140 TO 139
90 X = X1 * XS:Y = FN F(X)
100 IF 80 - Y < 0 OR 80 - Y > 160 THEN S = 0: GOTO 140
120 IF S = 0 THEN HPLLOT X1 + 140,80.5 - Y:S = 1: GOTO 140
130 HPLLOT TO X1 + 140,80.5 - Y
140 NEXT X1
150 PRINT "OK! THEN ";: GET G$: PRINT
155 IF G$ = CHR$(13) THEN 60
160 IF G$ = "N" THEN 25
165 IF G$ = "T" THEN 2
170 IF G$ = "E" THEN END
180 IF G$ = "B" THEN B = 1 - B:B$ = " B ": IF B = 0 THEN INPUT " FIX B
= ";G(2):B$ = STR$(G(2)): GOTO 275
190 IF G$ = "C" THEN C = 1 - C:C$ = " C ": IF C = 0 THEN INPUT " FIX C
= ";G(3):C$ = STR$(G(3)): GOTO 275
200 IF G$ = "A" THEN A = 1 - A:A$ = " A ": IF A = 0 THEN INPUT " FIX A
= ";G(1):A$ = STR$(G(1)): GOTO 275
203 IF TS < 3 THEN 275
205 IF G$ = "D" THEN D = 1 - D:D$ = " D ": IF D = 0 THEN INPUT " FIX D
= ";G(4):D$ = STR$(G(4))
275 TT = 8 * D + A * 4 + B * 2 + C
276 VTAB 21: IF TS > 5 THEN GOSUB 432
277 IF TS = 1 THEN GOSUB 315
278 IF TS > 2 AND TS < 6 THEN GOSUB 435
279 IF TS = 2 THEN GOSUB 312
281 VTAB 23
283 GOTO 150
285 FOR I = 1 TO 7: HPLLOT 137,I * 20 TO 143,I * 20: NEXT I
287 FOR I = 1 TO 13: HPLLOT I * 20,77 TO I * 20,83: NEXT I
300 XS = .05: IF G$ < > "N" THEN A = 1:TT = 7
312 IF TS = 2 THEN PRINT "Y=" SPC(4 - LEN (A$));A$*ABS(X-" SPC(5 -
LEN (B$));B$")+ " SPC(5 - LEN (C$));C$" V=1 H=1 *": RETURN
315 PRINT "Y=" SPC(5 - LEN (A$));A$*(X-" SPC(5 - LEN (B$));B$")^2
+" SPC(5 - LEN (C$));C$" V=1 H=1 *"
320 RETURN
420 FOR I = 0 TO 7:II = 80.5 - I * 11.14:IJ = 80.5 + 11.14 * I: HPLLOT 13
7,II TO 143,II: HPLLOT 137,IJ TO 143,IJ: NEXT I
422 IF TS = 3 THEN F$ = "SIN"
423 IF TS = 4 THEN F$ = "TAN"
425 IF G$ < > "N" THEN TT = 15:A = 1:D = 1
427 IF TS = 5 THEN F$ = "SEC"
430 FOR I = 1 TO 7: HPLLOT I * 35,77 TO I * 35,83: NEXT I
431 XS = 3.141592 / 35:YS = 1 / XS
432 IF TS = 6 THEN PRINT "Y=" SPC(5 - LEN (D$));D$*SIN(" SPC(5 - LEN
(A$));A$*X)+" SPC(5 - LEN (B$));B$*COS(" SPC(5 - LEN (C$));C$*
X)": RETURN
433 IF TS = 7 THEN PRINT "Y=" SPC(5 - LEN (D$));D$*SIN(" SPC(5 - LEN
(A$));A$*X)+" SPC(5 - LEN (B$));B$*SIN(" SPC(5 - LEN (C$));C$*
X)": RETURN
435 PRINT "Y=" SPC(4 - LEN (D$));D$*"F$(" SPC(4 - LEN (A$))A$*(X-
" SPC(4 - LEN (B$));B$))+ " SPC(4 - LEN (C$));C$" V=1 H=PI"
440 RETURN
1000 IF S0 = 0 THEN PRINT : PRINT CHR$(4)"BLOAD PAR,A$1E00"
1010 POKE 232,0: POKE 233,30
1020 R0 = 0:S0 = 1:TS = TS - 6:SE = 2:X$ = " X":Y$ = " Y"

```

```

1030 S1 = 0:S2 = 0:S3 = 0:S4 = 0:S5 = 0
1035 X2 = 140:Y2 = 80:X3 = 140:Y3 = 80:S1$ = "1600":S2$ = " 400"
1038 ROT= R0: SCALE= S0
1040 DRAW T5 AT X2,Y2
1042 HOME : VTAB 21
1044 IF TS = 7 THEN PRINT "  FUNCTION ";Y$;" = (" ;X$;"-";: INVERSE : PRINT
      SPC( 5 - LEN ( STR$ (S2)));S2$: NORMAL : PRINT ")^2 +";: INVERSE : PRINT
      SPC( 5 - LEN ( STR$ (S3)));S3: NORMAL
1045 IF TS = 8 THEN PRINT "ELLIPS (X-";: INVERSE : PRINT SPC( 4 - LEN
      ( STR$ (S4)));S4$: NORMAL : PRINT ")^2/"S1$"+(Y-";: INVERSE : PRINT SPC(
      4 - LEN ( STR$ (S5)));S5$: NORMAL : PRINT ")^2/"S2$="1";
1046 PRINT " [ I=UP,J=LEFT,K=RIGHT,M=DOWN,T=MENU ]"
1047 PRINT " [ 1-9=STEP,H=C C.W,L=C.W,E=END,N=NEW ]"
1048 POKE 34,23
1050 GET G$
1051 IF G$ = "N" THEN TS = TS + 1: GOTO 25
1052 IF G$ = "E" THEN TEXT : END
1053 IF G$ = "L" THEN R0 = R0 + 16
1054 IF G$ = "H" THEN R0 = R0 - 16: IF R0 < 0 THEN R0 = 256 + R0
1055 IF G$ = "T" THEN 2
1056 IF R0 = 256 THEN R0 = 0
1060 IF G$ = "I" THEN Y3 = Y2 - SE: IF Y3 < 0 THEN Y3 = 160
1070 IF G$ = "J" THEN X3 = X2 - SE: IF X3 < 0 THEN X3 = 278
1080 IF G$ = "K" THEN X3 = X2 + SE: IF X3 > 278 THEN X3 = 0
1090 IF G$ = "M" THEN Y3 = Y2 + SE: IF Y3 > 160 THEN Y3 = 0
1095 IF ASC (G$) > 48 AND ASC (G$) < 58 THEN SE = ASC (G$) - 48: GOTO
      1050
1100 HCOLOR= 0: DRAW T5 AT X2,Y2: HCOLOR= 3
1104 HPLOT 140,0 TO 140,160
1107 HPLOT 0,80 TO 279,80
1110 X2 = X3:Y2 = Y3
1120 R1 = R0 - INT (R0 / 64) * 64:R1 = R1 / 16
1125 S2 = (X2 - 140) / 20:S3 = (80 - Y2) / 20:S4 = X2 - 140:S5 = 80 - Y2
1130 IF R1 = 0 THEN X$ = " X":Y$ = " Y":S1$ = "1600":S2$ = " 400"
1135 IF R1 = 1 THEN X$ = " Y":Y$ = " X":TA = S2:S2 = S3:S3 = TA:S1$ = "
      400":S2$ = "1600"
1140 IF R1 = 2 THEN X$ = " X":Y$ = "-Y":S3 = - S3:S1$ = "1600":S2$ = "
      400"
1145 IF R1 = 3 THEN X$ = " Y":Y$ = "-X":TA = S2:S2 = S3:S3 = - TA:S1$ =
      " 400":S2$ = "1600"
1150 GOTO 1038

```

1

造型表(即程式行號 1000 中的 PAR)內容

ICALL-151	1E78- 24 2C 24 2C 24 2C 24 04
*1E00.1F05	1E80- 00 49 49 49 49 49 49 49
	1E88- 49 49 49 49 49 49 21 3C
	1E90- 24 3C 3C 24 27 3F 24 3F
1E00- 0A 03 18 00 81 00 06 01	1E98- 3C 3C 27 3F 3C 27 3F 27
1E08- 0D 01 2A 2A 2A 20 46 55	1EA0- 3F 3F 3C 3F 27 3F 3F 3F
1E10- 4E 43 54 49 4F 4E 20 59	1EAB- 27 3F 3F 3F 3F 3F 3F 37
1E18- 3F 3C 27 3F 27 27 27 27	1EB0- 3F 3F 3F 37 3F 3F 3E 3F
1E20- 27 27 3C 3C 3C 24 27 3C	1EB8- 37 3F 37 3F 3E 37 3F 3E
1E28- 24 27 3C 24 27 3C 24 27	1EC0- 3E 37 3E 37 37 3E 3E 36
1E30- 3C 24 27 24 27 3C 24 3C	1EC8- 3E 36 2E 36 2E 2E 36 35
1E38- 24 3C 24 3C 24 4C 49 49	1ED0- 2D 36 2D 2E 2E 35 2D 2E
1E40- 49 49 49 49 49 49 49 49	1ED8- 35 2D 35 2D 2D 2E 2D 35
1E48- 49 92 92 92 92 92 92 92	1EE0- 2D 2D 2D 35 2D 2D 2D 2D
1E50- 92 92 92 92 92 92 92 92	1EE8- 2D 2D 25 2D 2D 25 2D
1E58- DA DB 2D 2C 25 2D 25 25	1EFO- 2D 2C 2D 25 2D 25 2D 2C
1E60- 25 25 25 25 2C 2C 2C 24	1EF8- 25 2D 2C 2C 25 2C 25 25
1E68- 25 2C 24 25 2C 24 25 2C	1F00- 2C 2C 24 2C 24 00
1E70- 24 25 2C 24 25 24 25 2C	*

- 行號 2 ~ 19 呈現目錄頁。
- 行號 20 ~ 57 根據操作者之選擇，安置螢幕畫面及提示字幕。
- 行號 60 ~ 78 控制畫圖時，輸入數值之個數。
- 行號 80 ~ 86 控制選用的函數。
- 行號 87 ~ 140 在畫面上畫出函數圖形。
- 行號 150 ~ 283 控制操作者指令的程式流向。
- 行號 285 ~ 320 二次及絕對值函數畫圖時，安排坐標軸及提示字幕之副程式。
- 行號 420 ~ 440 三角函數畫圖時，安排坐標軸及提示字幕之副程式。
- 行號 1000 ~ 1048 呈現標準位置的拋物線或橢圓畫面及提示字幕。拋物線及橢圓是利用造型程式事先畫好的造型表。如附表所示。
- 行號 1050 ~ 1095 控制操作者指令的程式流向。
- 行號 1100 ~ 1150 平移或旋轉圖形，並修正所呈現的函數式或方程式。

本程式的特色，不在呈現函數的圖形，而在提供方便的改變係數值的方法，利用電腦，快速地畫出函數圖形，以便比較係數不同的函數圖形，幫助學生了解平移的意義及係數對函數圖形位置及形狀的影響。由於 BASIC 程式執行時速度較慢，若將程式略加修改，可用 TASC 編譯器編譯，這樣，執行速度會增快很多。當然，在行號 88 處，加上 STEP 2（或 3，4，5），亦可倍增畫圖速度，但這樣會破壞圖形的對稱性。另外，我們若想討論  $y = Df(A(x-B)) + C$  之圖形，只須將 80 至 86 行的函數定義改變即可（當然也要考慮  $f(x)$  所在的範圍。）。

關於此程式的應用，本文前段所提的一些問題，只是作者爲了補充課本的不足及幫助學生的學習而設計的，讀者在應用之時，可參酌課程內容，選用或自行設計合用的問題，來幫助學生。這種利用實驗來研究問題的方式，是相當值得推介的。